
Victor Vasarely et l’art optique

 L’Art optique rassemble des artistes fascinés par ce que l’œil perçoit du mouvement et de la lumière et 
qui en explorent les propriétés : intensité, rythme, variabilité, cycles... La démarche est bien sûr esthétique 
(non technique, physique ou métaphysique). C’est un mouvement riche et varié, qui s’exprime par des 
tableaux, des sculptures mais aussi des assemblages, installations ou montages vidéos. À ses débuts, on parle 
d’Art cinétique et optique. Dans les années 1960, l’appellation est simplifiée : on parle d’Art optique en 
France, d’Optique Art ou Op Art chez les anglo-saxons (d’après DYNAMO, le dossier pédagogique du Grand 
Palais).

  Victor Vasarely naît à Pécs en Hongrie en 1906. Il est un plasticien tout à fait singulier dans l’histoire 
de l’art du XXème siècle. Accédant à la notoriété de son vivant, il se distingue dans l’art contemporain par 
la création d’une nouvelle tendance : l’art optique. Ce mouvement s’attache à suggérer le mouvement sans 
jamais le réaliser véritablement. Il institue de nouvelles relations entre les spectateurs et l’œuvre en provo-
quant la participation active de celui qui regarde. Le spectateur est libre d’interpréter l’image en autant de 
situations visuelles qu’il pourra en concevoir. Son œuvre s’inscrit dans une grande cohérence, de l’évolution 
de son art graphique jusqu’à sa détermination pour promouvoir un art moins individualiste et plus commu-
nautaire, un art adapté aux mutations du monde moderne et du monde de l’industrie (d’après la biographie 
de la fondation Vasarely).

L’œuvre étudiée ici, qui s’intitule hexa 5, joue sur le regard du spectateur en proposant une illusion d’optique 
assez troublante. En effet, il suffit de regarder avec insistance le tableau ou de déplacer légèrement son point 
de vue pour que change radicalement la perception que l’on a de l’œuvre.

Étude géométrique

L’œuvre se compose de cinq hexagones presque réguliers. Chacun de ces hexagones est composé de nom-
breux losanges placés dans trois positions différentes qui permettent de donner l’impression de volume en 
faisant apparaître des cubes. Ces losanges ont donc obligatoirement des angles intérieurs de 60° et de 120°.
C’est l’irrégularité des hexagones ainsi que la coloration complexe de la figure géométrique qui permet l’illu-
sion d’optique quand nous regardons cette œuvre.

Éléments de réponse avec GéoTortue

Mise en place du problème

Le fait que les hexagones ne soient pas réguliers complique la tâche avec GéoTortue et oblige à bien penser en 
amont la structure du programme. Elle dépend aussi du niveau des élèves à qui l’on propose ce travail. En 
effet, cette structure peut être complètement différente selon que l’on fasse le choix de colorier manuellement 
avec l’interface de coloriage ou selon que l’on souhaite le faire par programmation. Dans le premier cas, le 
travail se limite à la construction de la figure géométrique de l’œuvre, le coloriage consiste ensuite en 690 
patients clics successifs. Dans le deuxième cas, l’utilisation de boucles et de conditions sont nécessaires ainsi 
qu’une structuration du programme bien particulière.
Il existe donc de nombreuses manières d’aborder le problème. Celle qui est proposée dans ces pages permet 
une mise en place plus aisée de la coloration par programmation.

Création du motif

Le motif de base est un losange d’angle intérieur de 60° et 120°. Nous choisissons ici de donner une longueur 
de 50 au côté de ce losange.

120°

60°

pour losange
rep 2 (av 50; td 120; av 50; td 60)
fin

Nous pouvons créer une procédure ligne, une procédure face et une procédure pavé qui nous serviront 
d’outils pour la construction et nous la rendront plus aisée.

pour pavé n h
face n h; lc; av h*50; td 60; av n*50; td 60; bc
face n n; lc ;av n*50; td 60; av h*50; td 60; bc
face h n; lc ;av n*50; td 60; av n*50; td 60; bc
fin

pour ligne n
rep n (losange; td 120; av 50; tg 120)
tg 60; lc; av nombre*50; bc ; td 60
fin

pour face n h
rep h (ligne n; av 50)
re h*50
fin

pour ligne nombre 
rep nombre (losange 50 120;td 120;av 50;tg 120)
tg 60;lc;av nombre*50;bc; td 60
fin

pour face_cube n h
rep h (ligne n;av 50)
re h*50
fin

Création d’un hexagone

La réalisation de l’hexagone peut-être facilitée par l’utilisation de la commande miroir qui permet de 
construire le symétrique d’une figure par rapport à une droite (en substituant les td par des tg et vice-versa) . 
Ainsi, nous créons une sous-procédure face_latérale qui nous permet de simplifier le programme.

pour face_intermediaire
face 3 3 
av 3*50
pavé 3 4
re 3*50
fin

Commande
miroir
face_intermediaire
miroir

Pour finir, après divers déplacements pour positionner la tortue correctement, nous pouvons assembler 
l’hexagone et obtenons la procédure hexagone suivante :

pour hexagone
face_intermediaire
lc; td 120; av 3*50; tg 120; bc
pavé 3 3
lc; td 120; av 3*50; tg 60; av 6*50; tg 60; bc
miroir; face_intermediaire; miroir
lc; av 7*50; tg 60; av 3*50; tg 60; av 3*50; td 60; bc
face 3 3
lc;av 3*50; td 120; re 3*50; tg 60; re 7*50; bc
fin

Création de la figure géométrique de l’œuvre

La construction de la figure finale est assez aisée. Seuls les deux hexagones disposés à l’envers demandent une 
attention particulière et quelques essais. Nous obtenons la procédure suivante :

pour figure
hexagone
lc; td 120; av 6*50; tg 60; av 6*50; tg 60; bc
hexagone
lc; av 7*50; tg 60; av 6*50; td 60; bc
hexagone
lc; av 7*50; td 60; av 6*50; td 60; av 6*50; tg 60; av 6*50; td 60; av 6*50; td 60; bc
hexagone
lc; av 7*50; td 60; av 6*50; td 60; av 6*50; tg 60; av 6*50; td 60; av 6*50; td 60; av 7*50; td 180; bc
hexagone
lc; av 7*50; td 60; av 6*50; tg 60; av 7*50; td 180
fin

Programmation des couleurs avec GéoTortue

Il y a deux couleurs par hexagone qui sont dégradées dans trois tons différents. Il y a toujours une couleur en 
commun entre certains hexagones quand on suit ce sens de lecture :

Cela nous donne six couleurs principales (vert-
bleu-orange-violet-mauve-rouge) dégradées dans 
trois tons différents (sombre-normal-clair) soit 
dix-huit couleurs au total.

Afin d’utiliser ces couleurs dans les procédures, 
nous créeons une procédure couleur qui va 
nous permettre d’affecter un numéro à cha-
cune des 18 couleurs. Les couleurs proposées 
sont celles qui se rapprochent le plus des cou-
leurs de l’œuvre.

pour couleur n
//bleu
si (n==1) alors (palette RVB 27 70 151)//sombre
si (n==2) alors (palette RVB 67 139 202)//normal
si (n==3) alors (palette RVB 100 178 228)//clair
//vert
si (n==4) alors (palette RVB 20 90 87)//sombre
si (n==5) alors (palette RVB 29 150 104)//normal
si (n==6) alors (palette RVB 69 176 106)//clair
//orange
si (n==7) alors (palette RVB 89 48 30)//sombre
si (n==8) alors (palette RVB 154 94 56)//normal
si (n==9) alors (palette RVB 197 137 53 )//clair
//violet
si (n==10) alors (palette RVB 64 49 118)//sombre
si (n==11) alors (palette RVB 93 84 160)//normal
si (n==12) alors (palette RVB 154 144 197)//clair
//mauve
si (n==13) alors (palette RVB 86 49 100)//sombre
si (n==14) alors (palette RVB 163 58 131)//normal
si (n==15) alors (palette RVB 227 83 152)//clair
//rouge
si (n==16) alors (palette RVB 89 48 30)//sombre
si (n==17) alors (palette RVB 164 26 22)//normal
si (n==18) alors (palette RVB 232 69 48)//clair
fin

Désormais, il faut adapter et modifier tout le programme afin qu’il intègre la gestion des couleurs.
Nous commençons par la procédure ligne dans laquelle nous devons alterner deux couleurs à chaque nou-
velle construction de losange. Nous devons donc affecter les couleurs en fonction de la parité de l’itération.
Ainsi, nous devons ajouter deux variables de couleurs (c1 et c2) à la procédure et utiliser une condition du 
type : si (reste(i,2)==0) alors (couleur c1) sinon (couleur c2), i représentant la position du losange.

Ensuite, deux possibilités s’offrent à nous pour repérer la position des losanges, qui auront les mêmes effets : 
créer un compteur d’itération interne à la commande rep ou utiliser la commande boucle.

Obtenu avec la commande :
 ligne n 5 8

pour ligne n c1 c2
compteur:=1
rep n (
si (reste(compteur,2)==0) alors (couleur c1) sinon (couleur c2)
remplis losange
lc; td 120; av 50; tg 120; bc
compteur:=compteur+1
)
lc; tg 60; lc; av n*50; bc; td 60; bc
fin

pour ligne n c1 c2
boucle i de 1 à n (
si (reste(i,2)==0) alors (couleur c1) sinon (couleur c2)
remplis losange
lc; td 120; av 50; tg 120; bc
)
lc; tg 60; lc; av n*50; bc; td 60; bc
fin

ou

Pour la suite, nous optons pour la procédure utilisant la commande boucle.

La procédure face se modifie de la même manière. L’affectation des couleurs s’alternant d’une ligne à 
l’autre, nous utilisons le même procédé que précédemment. Ce qui nous donne :

pour face n h c1 c2
boucle i de 1 à h (
si (reste(i,2)==0) alors (ligne n c1 c2) sinon (ligne n c2 c1)
lc; av 50; bc
)
lc; re h*50; bc
fin

Obtenu avec la commande :
 face 4 3 5 8

Pour la procédure pavé et la procédure face_intermédiaire, nous devons affecter 6 variables de couleur. 
De nombreux tests seront nécessaires pour placer les variables de couleur dans l’ordre lors de l’appel des 
différentes sous-procédures.

pour pave n h c1 c2 c3 c4 c5 c6
face n h c1 c2; lc; av h*50; td 60; av n*50; td 60; bc
face n n c3 c4; lc; av n*50; td 60; av h*50; td 60; bc
face h n c5 c6; lc; av n*50; td 60; av n*50; td 60; bc
fin

pour face_intermediaire c1 c2 c3 c4 c5 c6
face 3 3 c1 c2
lc;av 3*50; bc
pave 3 4 c2 c1 c3 c4 c5 c6
lc; re 3*50; bc
fin

La procédure hexagone est modifiée en conséquence :

pour hexagone c1 c2 c3 c4 c5 c6
face_intermediaire c1 c2 c3 c4 c5 c6
lc; td 120; av 3*50; tg 120; bc
pave 3 3 c2 c1 c4 c3 c5 c6
lc; td 120; av 3*50; tg 60; av 6*50; tg 60; bc
miroir; face_intermediaire c6 c5 c3 c4 c2 c1; miroir
lc; av 7*50; tg 60; av 3*50; tg 60; av 3*50; td 60; bc
face 3 3 c4 c3
lc; av 3*50; td 120; re 3*50; tg 60; re 7*50; bc
fin

De la même manière, après avoir créé une procédure pour le fond noir, la procédure figure est modifiée de la 
manière suivante : 

pour figure
fond
hexagone 7 4 9 6 8 5
lc; td 120; av 6*50; tg 60; av 6*50; tg 60; bc
hexagone 1 4 3 6 2 5
lc; av 7*50; tg 60; av 6*50; td 60; bc
hexagone 13 10 15 12 14 11
lc; av 7*50; td 60; av 6*50; td 60; av 6*50; tg 60; av 6*50; td 60; av 6*50; td 60; bc
hexagone 2 11 1 10 12 3
lc; av 7*50; td 60; av 6*50; td 60; av 6*50; tg 60; av 6*50; td 60; av 6*50; td 60; av 7*50; td 180; bc
hexagone 18 15 16 13 17 14
lc; av 7*50; td 60; av 6*50; tg 60; av 7*50; td 180; bc
fin

pour fond
lc; re 2000; td 90; re 2000; tg 90; bc
crayon noir
remplis rep 4 (av 4000; td 90)
lc; av 2000; td 90; av 2000; tg 90; bc
fin

Avec les élèves

Nous avons vu en étudiant le problème en détail que les exigences en terme de programmation dépen-
daient principalement du choix du mode de coloration de la figure géométrique. Sans colorier par pro-
grammation, ce problème est abordable dès la classe de 6è avec des élèves ayant une bonne maîtrise de 
GéoTortue. Les conditions et les boucles nécessaires à la programmation des couleurs sont assez basiques 
et peuvent être abordées à partir de la classe de 4è.

Cette œuvre peut aussi être proposée dans le cadre de l’histoire des arts. Il n’est nullement obligatoire de 
réaliser l’ensemble du programme Géotortue. Celui-ci peut d’ailleurs être utilisé par le professeur afin 
d’illustrer la construction de la figure géométrique en appelant petit à petit les différentes procédures 
composant le tableau final. Les élèves auront ainsi une idée du travail de conception réalisé par l’artiste.
Nous pouvons aussi aborder cette œuvre sur d’autres supports comme le papier pointé ou à l’aide du logi-
ciel Géogebra (en utilisant des macros) même si nous nous heurterons toujours au problème de la mise 
en couleur qui restera longue et fastidieuse. 

Pour aller plus loin

D’autres réalisations sont possibles à partir des procédures écrites pour reproduire cette œuvre. Il existe 
d’autres œuvres de Victor Vasarely qui reprenne le même principe d’illusion d’optique. Nous pouvons aussi 
laisser la liberté aux élèves de créer et d’imaginer leurs propres figures.

Gestalt 4 (1970) Bi-Hexa (1975)

